

4.exportData

March 3, 2025

0.0.1 Exporting Data

In Exporting Results you learn to export SAS tables and results to Excel, Microsoft Word, and PDF files.

Exporting Data

- PROC EXPORT can export a SAS table to a variety of file formats outside SAS.

```
PROC EXPORT DATA=input-table OUTFILE="output-file"  
             <DBMS=identifier> <REPLACE>;  
RUN;
```

- The XLSX engine requires a license for SAS/ACCESS to PC Files.
- The XLSX engine can read and write data in Excel files.
- To write data to a new or existing Excel workbook, use the LIBNAME statement to assign a libref pointing to the Excel file. Use the libref when naming output tables. The table name is the worksheet label in the Excel file.

```
OPTIONS VALIDVARNAME=V7;
```

```
LIBNAME libref XLSX "path/file.xlsx";
```

- Use libref for output table(s).

```
LIBNAME libref CLEAR;
```

```

/*exporting sas table as a csv */
*using a macro variable;
proc export data=pg1.storm_final outfile="&outpath/storm_final.csv" dbms=csv replace ;
run;

/*exporting sas table as a excel */

libname xout xlsx "&outpath/southPacific.xlsx";
data xout.South_Pacific;
    set pg1.storm_final;
    where Basin="SP";
run;

proc means data=pg1.storm_final noprint maxdec=1;
    where Basin="SP";
    var MaxWindKM;
    class Season;
    ways 1;
    output out=xout.Season_Stats n=Count mean=AvgMaxWindKM max=StrongestWindKM;
run;
libname xout clear;

```

0.0.2 Exporting Reports

Exporting Reports

- The SAS Output Delivery System (ODS) is programmable and flexible, making it simple to automate the entire process of exporting reports to other formats.

```

ODS <destination> <destination-specifications>;

/* SAS code that produces output */

ODS <destination> CLOSE;

```

- Selected destinations:
 - CSVALL
 - PowerPoint
 - RTF
 - PDF

```

ods csvall file="&outpath/cars.csv";
proc print data=sashelp.cars noobs;
    var Make Model Type MSRP MPG_City MPG_Highway;
    format MSRP dollar8.;
run;
ods csvall close;

```

```

ODS CSVALL FILE="filename.csv";
/* SAS code that produces output */
ODS CSVALL CLOSE;

```

```
ODS POWERPOINT FILE="filename.pptx" STYLE=style;  
/* SAS code that produces output */  
ODS POWERPOINT CLOSE;
```



```
ODS RTF FILE="filename.rtf" STARTPAGE=NO;  
/* SAS code that produces output */  
ODS RTF CLOSE;
```



- Exporting results to Excel:

```
ODS EXCEL FILE="filename.xlsx" STYLE=style  
OPTIONS(SHEET_NAME='label');  
  
/* SAS code that produces output */  
  
ODS EXCEL OPTIONS(SHEET_NAME='label');  
  
/* SAS code that produces output */  
  
ODS EXCEL CLOSE;
```

- The ODS EXCEL destination creates an .XLSX file.
- By default, each procedure output is written to a separate worksheet with a default worksheet name. The default style is also applied.
- Use the STYLE= option on the ODS EXCEL statement to apply a different style.
- Use the OPTIONS(SHEET_NAME='label') on the ODS EXCEL statement to provide a custom label for each worksheet.
- Exporting results to PDF:

```
ODS PDF FILE="filename.pdf" STYLE=style  
STARTPAGE=NO PDFTOC=1;  
ODS PROCLABEL"label";  
  
/* SAS code that produces output */  
  
ODS PDF CLOSE;
```

- The ODS PDF destination creates a .PDF file.
- The PDFTOC=n option controls the level of the expansion of the table of contents in PDF documents.
- The ODS PROCLABEL statement enables you to change a procedure label.